// File: Splitter_Stencils.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.16;

import { IERC20 } from "@openzeppelin/contracts/token/ERC20/IERC20.sol";

contract Splitter {

    uint256 internal constant IS_NOT_LOCKED = uint256(1);
    uint256 internal constant IS_LOCKED = uint256(2);

    uint256 internal _lockedStatus = IS_NOT_LOCKED;

    address[2] public accounts;

    mapping (address => uint256[3]) public banked;

    constructor(address[2] memory accounts_) {
        accounts[0] = accounts_[0];
        accounts[1] = accounts_[1];
    }

    modifier noReenter() {
        require(_lockedStatus == IS_NOT_LOCKED, "NO_REENTER");

        _lockedStatus = IS_LOCKED;
        _;
        _lockedStatus = IS_NOT_LOCKED;
    }

    function changeAccount(uint256 index_, address account_) external {
        require(accounts[index_] == msg.sender, "UNAUTHORIZED");
        accounts[index_] = account_;
    }

    function splitETH() external noReenter {
        uint256 totalBalance = address(this).balance;

        // NOTE: we don't care about success. If it fails, that account loses out on that portion.
        //       Leftover funds will get re-split next call.
        _transferEth(accounts[0], (totalBalance * 85) / 100);
        _transferEth(accounts[1], (totalBalance * 15) / 100);
    }

    function splitToken(address token_) external noReenter {
        uint256 totalBalance = IERC20(token_).balanceOf(address(this));

        // NOTE: We don't care about success. If it fails, that account loses out on that portion.
        //       Leftover funds will get re-split next call.
        IERC20(token_).transfer(accounts[0], (totalBalance * 85) / 100);
        IERC20(token_).transfer(accounts[1], (totalBalance * 15) / 100);
    }

    function _transferEth(address destination_, uint256 amount_) internal returns (bool success_) {
        ( success_, ) = destination_.call{value: amount_}("");
    }

    receive() payable external {}

    fallback() payable external {}

}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}